```
(*this Mathematica notebook evaluates several
 observables of the 1D Ising model and plots them
 in the temperature/field plane, derivatives (of the free energy)
 are evaluated using the Mathematica functions
 D[f[x],x] for df/dx and Derivative[0,2][f[x,y]] for d^2f/dy^2 *)

(*1D Ising model with J=1,hence Jtilde=1/T=b *)

(*the two eigenvalues of the transfer matrix as a
 function of inverse temperature b and magnetic field*)
lambda1[b_, h_] := Exp[b] Cosh[b h] + Exp[b] Sqrt[Sinh[b h]^2 + Exp[-4 b]]
lambda2[b_, h_] := Exp[b] Cosh[b h] - Exp[b] Sqrt[Sinh[b h]^2 + Exp[-4 b]]
```

$$e^{b}\,\mathrm{Cosh}[b\,h] + e^{b}\sqrt{e^{-4\,b} + \mathrm{Sinh}[b\,h]^{2}}$$

$$e^{b}\,\mathrm{Cosh}[b\,h] - e^{b}\sqrt{e^{-4\,b} + \mathrm{Sinh}[b\,h]^{2}}$$

```
(*(ln Z)/N is equal the logarithm of the maximum eigenvalue,
defined here explicitly*)
loglambda1[b_, h_] := Log[lambda1[b, h]]

(*magnetisation evaluated by taking the derivative of lnZ wrt to h*)
m[b_, h_] = 1/b D[Log[lambda1[b, h]], h]
```

$$\frac{b\,e^{b}\,\mathrm{Sinh}[b\,h] + \frac{b\,e^{b}\,\mathrm{Cosh}[b\,h]\,\mathrm{Sinh}[b\,h]}{\sqrt{e^{-4\,b}+\mathrm{Sinh}[b\,h]^{2}}}}{b\left(e^{b}\,\mathrm{Cosh}[b\,h] + e^{b}\sqrt{e^{-4\,b} + \mathrm{Sinh}[b\,h]^{2}}\right)}$$

```
(*magnetisation plotted in the T,h plane*)
Plot3D[m[1/T, h], {T, 0, 5}, {h, -2, 2}]
```



```
(* expected energy in t,h plane*)
energy[b_, h_] := -D[Log[lambda1[b, h]], b]
```

$$-\frac{e^{b}\,\text{Cosh}[b\,h]+e^{b}\,h\,\text{Sinh}[b\,h]+\frac{e^{b}\left(-4\,e^{-4\,b}+2\,h\,\text{Cosh}[b\,h]\,\text{Sinh}[b\,h]\right)}{2\,\sqrt{e^{-4\,b}+\text{Sinh}[b\,h]^{2}}}+e^{b}\,\sqrt{e^{-4\,b}+\text{Sinh}[b\,h]^{2}}}{e^{b}\,\text{Cosh}[b\,h]+e^{b}\,\sqrt{e^{-4\,b}+\text{Sinh}[b\,h]^{2}}}$$

```
Plot3D[energy[1/T, h], {T, 0, 5}, {h, -2, 2}]
```




```
(* - beta f = Log[lambda1]. plot f free energy in t,h plane*)
Plot3D[-T Log[lambda1[1/T, h]], {T, 0, 5}, {h, -2, 2}]
```




```
(* s=ln Z +beta e. entropy in t,h plane*)
(*the entropy per spin is ln2 at high temperatures,
but reaches zero at low temperatures*)
```

```
Plot3D[Log[lambda1[1/T, h]] + 1/T energy[1/T, h], {T, 0, 5}, {h, -2, 2}]
```




```
(*the magnetic susceptibility dm/dh is second derivative of lnZ wrt h*)
susceptibility[b_, h_] := 1/b^2 Derivative[0, 2][loglambda1][b, h]
```

```
(*note the divergence of the magnetic susceptibility at h=
 0 as the temperature is lowered*)
Plot3D[susceptibility[1/T, h] , {T, 0, 5}, {h, -2, 2}]
```




```
(*the specific heat is second derivative of lnZ wrt beta*)
specificheat[b_, h_] := b^2 Derivative[2, 0][loglambda1][b, h]
```

```
Plot3D[specificheat[1/T, h] , {T, 0, 5}, {h, -2, 2}]
```




```
(*the difference between the two
 eigenvalues vanishes with decreasing temperature*)
Plot[{lambda1[1/T, 0], lambda2[1/T, 0]}, {T, 0, 5}]
```